

计算机实验指导丛书

BASIC

程序设计实验指导书

翁瑞琪 李树钰 编

天津大学出版社



计算机实验指导丛书

BASIC程序设计实验指导书

翁瑞琪 李树钰 编

天津大学出版社

TP312-45-

内 容 简 介

本书是为满足BASIC程序设计实验课需要而编写的实验指导书,是《计算机实验指导丛书》的一个分册。

书中包括BASIC上机操作初步、输入与输出、分支、循环、标准函数及自定义函数、子程序、汉字的使用、数组的使用、程序的效率、字符串处理、顺序文件的处理、随机文件的处理等12个实验。对于每个实验,书中都提出明确的实验目的、预习要求、实验内容和步骤并有相当数量的实验习题。

书中附有IBM PC、APPLE II、COMX35、LASER-310、PZ-80、MC 68000、DPS8等7种机器的使用操作说明以及汉字使用说明。

本书可供高等院校、电大、职大学生使用,也可供在职人员自学BASIC程序设计时实习使用。

计算机实验指导丛书 BASIC程序设计实验指导书

翁瑞琪 李树钰 编

*

天津大学出版社出版

(天津大学内)

河北省永清县印刷厂印刷

新华书店天津发行所发行

*

开本: 787×1092毫米 $1/16$ 印张: $6\frac{1}{4}$ 字数: 150千字

1988年9月第一版 1988年9月第一次印刷

印数: 1—10000

ISBN 7-5618-0061-4

TP·9

定价: 1.10元

编 者 的 话

实验是掌握科学技术的重要手段。特别是，要掌握计算机这门实践性很强的学科，上机实验是学习过程不可缺少的环节。通过上机实验，可以加深对计算机基本概念的理解，又可培养使用计算机的能力。

计算机是实现现代化的重要手段。其应用已深入到国民经济的各个领域。计算机及其应用方面的课程现已成为我国高等院校大多数专业的必修课。已开设了《计算机程序设计》（如BASIC程序设计、FORTRAN程序设计、COBOL程序设计等）和《微机原理及其应用》等课程。大多数高等院校还陆续配置了各种类型计算机，可以提供较充裕的上机机时。在此形势下，迫切需要一套适用的计算机实验指导丛书。有鉴于此，我们组织天津大学计算机系、计算中心、技术经济与系统工程系长期从事计算机教学和实验指导工作的教师联合编写了本丛书，由翁瑞琪副教授任主编。

本丛书预定包括以下五个分册：

- （1）《单板计算机实验指导书》；
- （2）《FORTRAN程序设计实验指导书》；
- （3）《BASIC程序设计实验指导书》；
- （4）《COBOL程序设计实验指导书》；
- （5）《dBASE的使用与操作实验指导书》。

即将陆续出版。

前 言

目前, BASIC程序设计 (BASIC语言) 课已被高等院校的许多专业列为必修课程。为满足这一课程开设的需要, 国内已出版了多种BASIC程序设计 (BASIC语言) 教材。但相应的实验指导书却比较缺乏。上机实验是掌握BASIC程序设计课的重要一环。为了适应教学需要, 我们编写了本实验指导书。

本书包括上机操作初步、输入与输出、分支、循环、函数、子程序、汉字、数组、程序效率、字符串、磁盘文件操作等12个实验。这些实验旨在加深学生对BASIC程序设计的基本概念的理解、提高他们使用操作机器的能力以及培养他们编写程序和调试程序的能力。

对于每个实验, 都提出了明确的实验目的和预习要求。预习要求中除给出实验前应预习的内容外, 有的还给出问题让学生自己编写实验程序。学生通过对自编程序的调试将会有较大的收获。在实验内容和步骤中给出的实验量较大, 这样有利于教师根据具体情况作适当节选。实验习题中留有相当数量的习题。有的习题与实验内容紧密结合, 用来让学生在实验后进行思考和小结; 有的习题用来补充实验内容上的不足。为了达到预期效果, 望教师严格掌握以下几点:

(1) 凡未按要求作好预习的学生不得上机实验。

(2) 强调学生独立地去做实验 (尤其是程序的调试)。

(3) 实验后要求学生必须认真地书写实验报告 (包括完成给定的实验习题)。

本书可作为高等院校各专业BASIC程序设计 (BASIC语言) 课的实验用书。也可作为电视大学、职工大学、函授大学BASIC实验指导用书, 还可供在职人员自学BASIC语言作实验时使用。

本书给出的每个实验约需2~3学时, 12个实验的上机学时数合计为24~36。鉴于各专业要求的上机学时数有所不同, 各校的上机条件也有所不同, 因此, 12个实验中有4个标有*号, 对其可根据情况选做。

鉴于各校配备的机种有所不同, 本书附录给出了IBM PC、APPLE II、COMX 35、LASER-310、PZ-80、MC68000、DPS8/52等7种机器的使用与操作方法, 因此, 本书有较大的适应性。

要说明的是, 各种机器的BASIC版本在BASIC基本语句方面的基本规定是一致的, 但扩展情况和操作命令不一定完全相同, 本书无法做得面面俱到。鉴于国内目前配备IBM PC和其兼容机的实验室较多, 因此实验中有些内容侧重于以IBM PC为典型, 望采用其他机种进行实验的单位加以注意。

本书由天津大学技术经济与系统工程系翁瑞琪、李树钰合编。

本实验指导书是《计算机实验指导丛书》的一个分册。

目 录

实验一	上机操作初步.....	(1)
实验二	输入与输出.....	(4)
实验三	分支.....	(7)
实验四	循环.....	(10)
实验五	标准数值函数及自定义函数.....	(13)
实验六	子程序.....	(18)
* 实验七	汉字的使用.....	(21)
实验八	数组的使用.....	(30)
实验九	程序的效率.....	(33)
* 实验十	字符串的处理.....	(38)
* 实验十一	磁盘文件操作 (一)	(41)
* 实验十二	磁盘文件操作 (二)	(44)
附录 1	IBM PC机的使用与操作.....	(46)
附录 2	APPLE II 机的使用与操作.....	(59)
附录 3	COMX 35机的使用与操作.....	(66)
附录 4	LASER-310机的使用与操作.....	(76)
附录 5	PZ-80机的使用与操作.....	(80)
附录 6	MC 68000机的使用与操作.....	(83)
附录 7	DPS8/50机的使用与操作.....	(87)

实验一 上机操作初步

【实验目的】

1. 了解键盘的配置并熟悉键盘的操作;
2. 掌握启动BASIC的过程;
3. 掌握NEW、RUN、LIST等BASIC命令的使用;
4. 掌握如何输入、编辑和运行程序。

【预习要求】

1. 根据实验使用机器的型号, 阅读附录 (或有关说明书) 中给出的该种型号机器的使用与操作方法;

2. 预习BASIC程序的构成和基本规则方面的内容。

【实验内容和步骤】

1. 使机器进入BASIC状态。
2. 键入NEW命令。
3. 键入下列“计算平均值和标准偏差”的程序:

```
10 REM CALCULATE MEAN AND STANDARD DEVIATION
20 LET S1 = 0
30 LET S2 = 0
40 LET N = 0
50 READ X
60 IF X = 0 THEN 110
70 LET S1 = S1 + X
80 LET S2 = S2 + X * X
90 LET N = N + 1
100 GOTO 50
110 LET M = S1 / N
120 LET V = S2 / N - M * M
130 LET S = SQR(V)
140 PRINT "MEAN = "; M, "STANDARD DEVIATION = "; S
150 DATA 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 0
160 END
```

4. 用LIST命令在屏幕上列出刚键入的程序, 查看有无键错之处。如果有错的话, 利用屏幕进行编辑, 直到正确为止。

5. 若程序键入无错或已改正, 则用RUN命令运行上述程序, 根据显示的运行结果估算程序运行是否正常。

6. 若运行正常, 把程序中的PRINT加一个L而成为LPRINT, 用LLIST命令打印程

序清单并用RUN命令再运行此程序而把运行结果也打印出来。

7. 将程序作如下修改:

1) 把10号语句中的AND STANDARD DEVIATION等字符删去。

2) 把30号、80号、120号、130号等语句从程序中删去。

3) 把140号语句改为: 140 PRINT "MEAN = ", M

4) 把150号语句中最后一个逗号和其后的0删去。

5) 增添155号语句如下:

155 DATA 11, 12, 13, 14, 15, 16, 17, 18, 19, 20, 0

8. 完成以上修改后用LIST命令查看是否修改正确。若修改无误, 用RUN命令运行修改过的程序。估算程序运行是否正常。

9. 若正常的话, 用同前的方法打印程序清单和运行结果。

[实验习题]

1. 在运行一程序后未键入NEW命令就键入一个新程序, 将会出现什么情况?

例如, 原有下列程序:

10 LET A = 1.2

20 LET B = 2.3

30 LET C = 3.4

40 PRINT A + B + C

50 END

运行后得到运行结果为6.9, 在未键入NEW命令情况下现又键入下列程序:

5 LET A = 1.1

10 LET B = 2.2

15 LET C = 3.3

20 LET D = 4.4

35 PRINT A + B

45 PRINT C + D

55 PRINT A + B + C - D

65 END

试问机器存储器中现有程序是什么? 其运行结果将会是什么?

2. 行号和回车符各起什么作用?

假设存储器中现有程序如下:

10 LET A = 123

20 LET B = 456

30 PRINT A + B

40 PRINT B - A

50 END

现要删除其中的30号语句。

1) 学生甲在屏幕上擦除30号语句中的PRINT A + B等字符, 但擦去后未按回车键。试问30号语句是否已从程序中删去? 为什么?

2) 学生乙在屏幕上把30 PRINT A + B等字符全部擦去, 且擦去后按了一下回车键。

试问30号语句是否已从程序中删去? 为什么?

3) 删去30号语句的正确方法有哪几种?

3. 假设学生丙在键入NEW命令后键入一个新程序, 但是他在未键入RUN命令的情况下就试图把10.25这个数赋给X, 即他在键完程序后就立即键入10.25并按下回车键。于是, 屏幕上出现如下显示:

```
10 INPUT X
20 IF X < 0 THEN 50
30 PRINT "SQR(X) = "; SQR(X)
40 GOTO 60
50 PRINT "SQR(-X) = "; SQR(-X)
60 END
10.25
```

试问在此情况下键入LIST命令, 则屏幕上将列出怎样的程序清单? 为什么?

实验二 输入与输出

[实验目的]

1. 掌握LET语句的赋值与运算功能及其正确用法;
2. 掌握键盘输入语句INPUT的正确用法;
3. 掌握READ、DATA与RESTORE语句的配合使用;
4. 掌握PRINT语句的功能及其输出格式;
5. 进一步熟悉如何编辑程序和初步学习如何调试程序。

[预习要求]

1. 预习LET语句的语法规定;
2. 预习INPUT语句的语法规定;
3. 预习READ、DATA和RESTORE语句的语法规定;
4. 预习PRINT语句的语法规定。

[实验内容和步骤]

1. 照样输入下列程序并试运行。

```
10 LET A = 1/2
20 LET B = 3
30 LET C = 1/4
40 LET D = B2 - 4AC
50 PRINT "AX2 + BX + C = 0"
60 PRINT "A = ", A, "B = ", B, "C = ", C
70 PRINT X1 = (-B + SQR(D))/(2 * A)
80 PRINT X2 = (-B - SQR(D))/(2 * A)
90 END
```

上列程序是求方程 $ax^2 + bx + c = 0$ 的两个实根的程序,但它是一个含有语法错误的程序,因此计算机在运行时将给出错误信息。试根据错误信息进行修改(调试),直至正常运行。在运行正常后,把程序中的PRINT改为LPRINT,打印程序清单和运行结果。

由运行结果可知,上述程序用来求 $0.5x^2 + 3x + 0.25 = 0$ 的两个实根。如果现在要利用此程序求 $2x^2 + 7x + 4 = 0$ 的两个实根,试按此要求修改上述程序并运行之。

2. 对现有程序作如下修改:

- 1) 把10号语句改为

```
10 PRINT "A, B, C = ";
```

- 2) 把20号语句改为

```
20 INPUT A, B, C
```

- 3) 把30号语句删去。

- 4) 把50号、60号、70号、80号语句中的LPRINT改为PRINT。

作上述修改后试运行此程序(运行时通过键盘输入给A, B, C的数据为0.5, 3, 0.25)。

由于现在20号语句有语法错误，运行时计算机将给出错误信息。试根据错误信息进行调试，直至正常运行。运行正常后，把50号、60号、70号、80号语句中的PRINT改为LPRINT，打印出程序清单。然后运行该程序5次，使之打印出运行结果。各次运行时通过键盘分配给A, B, C的数据如下：

1) 0.5, 3, 0.25

2) 2, 7, 4

3) 3, 8, 1

4) 2.5, 6.1, 3.2

5) 2.3, 3.4, -1.2

3. (1)对现有程序再作如下修改：

1) 把10号语句改为

10 DATA 1/2, 3, 1/4

2) 把20号语句改为

20 READ A, B, C

3) 把50号、60号、70号、80号语句中的LPRINT改为PRINT。

作上述修改后试运行并调试此程序。调试正常后，把50号、60号、70号、80号语句中的PRINT改为LPRINT。先打印出程序清单，然后运行程序使之打印运行结果。

(2) 对现有程序再作如下修改：

1) 把10号语句改为

10 DATA 0.5, 3, 0.25, 2, 7, 4, 3, 8, 1, 2.5, 6.1, 3.2, 2.3, 3.4,
-1.2

2) 增添85号语句如下

85 GOTO 20

作上述修改后，打印出程序清单，然后运行程序使之打印运行结果。

4. 键入下列新程序（注意：键入前先键入NEW命令）并试运行之。

10 DATA, 1, 2, 3, 4, 5, 6

20 READ A, B, C, D, E

30 PRINT A, B, C, D, E

40 RESTORE

50 READ A, A, B, B, B, C

60 PRINT A, B, C, D, E

70 END

若运行正常，设法打印程序清单和运行结果。

5. 键入下列新程序并试运行之。

10 LET A = -1.2

20 LET B = 2.3

30 PRINT A, B

40 PRINT A, B

50 PRINT "A + B = "; A, "+", B, "=", A + B

```
60 PRINT "A + B = ", "-1.2 + 2.3", "=", "1.1"
70 END
```

若运行正常，设法打印程序清单和运行结果。

[实验习题]

1. 对步骤 1、2、3 中给 A, B, C 提供数据的 3 种方法进行比较，分析它们各适于哪种使用情况。

2. 下面两个程序是否有同样的运行结果？为什么？

程序 1

```
10 LET A = 1.2
20 LET B = 2.3
30 LET A = B
40 PRINT A; B; A + B
50 END
```

程序 2

```
10 LET A = 1.2
20 LET B = 2.3
30 LET B = A
40 PRINT A; B; A + B
50 END
```

3. 步骤 3 (2) 中经修改后的程序运行时，除了得到运行结果外还输出什么信息？它说明什么问题？它何以发生？

4. 步骤 4 程序运行中 30 号语句输出的 A, B, C, D, E 值与 60 号语句输出的 A, B, C, D, E 值有哪些相同，哪些不同？为什么会这样？

5. 步骤 5 程序运行时，30 号语句的输出与 40 号语句的输出有何区别？为什么？50 号语句的输出与 60 号语句的输出有什么区别？为什么？

实验三 分 支

〔实验目的〕

1. 掌握如何运用IF语句实现条件分支;
2. 初步实践如何依据解题流程编写程序;
3. 初步学习如何设计测试用例;
4. 掌握调试程序的初步技能。

〔预习要求〕

1. 预习IF语句的语法规定和用法;
2. 预习流程图的画法;
3. 阅读本实验附注“设计测试用例简介”;
4. 根据下面给出的题意编制流程图, 然后按照流程图编写程序并设计测试用例。

题1 编写一个由键盘输入 x 值, 按下式计算并输出 y 值的程序:

$$\begin{cases} y = -1 & x \leq -1 \\ y = x & -1 < x < 1 \\ y = 1 & x \geq 1 \end{cases}$$

题2 编写由三角形三个边长计算三角形面积的程序。边长由键盘输入。若输入数据不正确(两边之和不大于第三边), 程序给出输入数据有错的信息并要求重新送数。

题3 对球队队员按下列方法分类进行统计:

第一类 $W \geq 60, L \geq 1.8$

第二类 $W < 60, L \geq 1.8$

第三类 $W \geq 60, L < 1.8$

第四类 $W < 60, L < 1.8$

其中 W 为体重(以kg为单位), L 为身长(以m为单位)。

编写统计各类队员人数和占总数比例的程序。队员的体重和身长由键盘输入语句提供。若键入的 W 为0或 L 为0时表示输入数据送完。

〔实验内容和步骤〕

1. 输入按题1自编的由 x 计算 y 值的程序并调试之。利用所设计的测试用例测试程序, 用以寻找错误。若找出错误, 便修改程序。

2. 输入按题2自编的计算三角形面积的程序并调试之。利用所设计的测试用例进行测试。

3. 现有的计算三角形面积的程序, 每运行一次, 求一个三角形的面积。现在, 对它进行修改。使之一旦运行就不断地计算各三角形的面积, 直至键入的数据为0, 0, 0时程序才终止执行。

4. 输入按题3自编的统计球队队员体型分类的程序并调试之。用所设计的测试用例进行测试。

[实验习题]

1. 在上述实验程序的调试中, 机器是否给出过什么错误信息? 这些错误信息的含义是什么? 何以会出现这样的错误信息? 你是怎样解决的?

2. 下列程序是计算产量在日增长率为 2%、3%、5% 的情况下, 增长 20% 所需天数的程序。

```
10 READ I
20 IF I = 0 THEN 110
30 LET Q = 1
40 LET D = 0
50 LET Q = Q(1 + I/100)
60 LET D = D + 1
70 IF Q < 1.2 THEN 50
80 PRINT "Increasing Rate = ", I, "%", "Days Required = ", D,
  "Practical Increment = ", (Q - 1) * 100, "%"
90 GOTO 10
100 DATA 2, 3, 5, 0
110 END
```

画出此程序的流程图。

3. 画出下列程序的流程图并写出当赋给 X 的数为 12.34 时该程序的运行顺序。

```
10 INPUT X
20 IF X < 0 THEN 50
30 PRINT "X NOT LESS THAN ZERO"
40 GOTO 60
50 PRINT "X LESS THAN ZERO"
60 END
```

把此程序改编成不带 GOTO 语句的程序, 并写出改编后的程序的运行顺序。

附注 设计测试用例简介

设计程序的目的是为了运用它。因此, 设计者理应保证程序无错。不仅要通过调试修正所有的语法错误, 还应在输入各种可能的数据时都能得到正确的结果。对小而简单的程序, 这是较易做到的。但是, 对于大而复杂的程序, 做到这一点并不容易。因此, 我们往往用把一些数据输入给程序, 看运行时能否得到预期输出的方法测试程序是否有错。测试的目的是为了发现错误。事实上, 不可能进行无限次测试。因此, 就要精心选择一些有代表性的数据组合 (尤其是容易暴露错误的数据组合) 进行测试。我们把测试前选择测试用数据和算出预期输出这样的工作叫做设计测试用例。例如, 对于实验习题 3 中的程序, 可设计测试用例如下:

例子	输 入	预期输出		
1	<table><tr><td>A</td><td>- 3</td></tr></table>	A	- 3	X LESS THAN LERO
A	- 3			
2	<table><tr><td>A</td><td>2.4</td></tr></table>	A	2.4	X NOT LESS THAN LERO
A	2.4			
3	<table><tr><td>A</td><td>0</td></tr></table>	A	0	X NOT LESS THAN LERO
A	0			

实验四 循 环

[实验目的]

1. 掌握单重循环和多重循环的用法;
2. 进一步提高编写程序和调试程序的能力;
3. 掌握把机器存储器中的程序存入磁盘 (或磁带) 以及把磁盘 (或磁带) 上的程序装入机器的方法。

[预习要求]

1. 预习 FOR- NEXT 语句;
2. 根据下面给出的题意编写程序:

题 1 采用单重循环编写计算 20 个数 ($a_1 \sim a_{20}$) 的 20 个平均值 ($m_1 \sim m_{20}$) 的程序。 $a_1 \sim a_{20}$ 由读数据语句提供。平均值 $m_1 \sim m_{20}$ 按下列算式计算:

$$m_1 = \frac{a_1 + a_2}{2}$$

$$m_i = \frac{a_{i-1} + a_i + a_{i+1}}{3} \quad (i = 2 \sim 19)$$

$$m_{20} = \frac{a_{19} + a_{20}}{2}$$

题 2 采用双重循环编写输出下列图案的程序。

```
>
>>
>>>
>>>>
>>>
>>
>
```

[实验内容和步骤]

1. 照原样输入下列九九表程序

```
10 FOR I=1 TO 9
20 PRINT I;
30 NEXT I
40 FOR I=1 TO 9
50 PRINT I
60 FOR J=1 TO 9
70 PRINT A*B;
80 NEXT I
```

90 NEXT J

100 END

此程序有若干错误。调试此程序使之输出如下运行结果：

	1	2	3	4	5	6	7	8	9
1	1	2	3	4	5	6	7	8	9
2	2	4	6	8	10	12	14	16	18
3	3	6	9	12	15	18	21	24	27
4	4	8	12	16	20	24	28	32	36
5	5	10	15	20	25	30	35	40	45
6	6	12	18	24	30	36	42	48	54
7	7	14	21	28	35	42	49	56	63
8	8	16	24	32	40	48	56	64	72
9	9	18	27	36	45	54	63	72	81

2. 输入按题1自编的求20个平均值的程序并调试之。

3. 输入按题2自编的输出图案的程序并调试之。

4. 下面示出的是一完整程序的一部分（子程序），它用来求 $C_M^N = \frac{N!}{M!(N-M)!}$

```
1000 REM SUB1: CALCULATE COEFFICIENT C(N,M)
```

```
1010 LET K = N
```

```
1020 GOSUB 1110
```

```
1030 LET C = P
```

```
1040 LET K = M
```

```
1050 GOSUB 1110
```

```
1060 LET C = C/P
```

```
1070 LET K = N - M
```

```
1080 GOSUB 1110
```

```
1090 LET C = C/P
```

```
1100 RETURN
```

```
1110 REM SUB2: CALCULATE FACTORIAL
```

```
1120 LET P = 1
```

```
1130 IF K = 0 THEN 1170
```

```
1140 FOR I = 1 TO K
```

```
1150 LET P = P * K
```

```
1160 NEXT I
```

```
1170 RETURN
```

先把它输入机器中，然后用SAVE命令把它作为程序文件存入磁盘（或磁带）。

接着用NEW命令清除机器存储器并用LIST命令查看是否确已清除。之后用LOAD命令把它又装入机器并用LIST命令查看是否确已装入。

这个子程序要保留下来，以备实验六中使用。